Mini-Wasm Tutorial

1 Introduction

In this tutorial, we will practice how to write a spec in Wasm-DSL. Here, instead of a full Wasm, a very simplified version of Wasm (which we call Mini-Wasm) is used as our goal.

1.1 Working Directory

The directory `<root>/spectec/tutorial` is where we will work on.

1.2 How To Run

Running `make` in `<root>/spectec` will yield an executable file `watsup` in same directory. You can use this to generate various types of spec. For example, in directory `<root>/spectec`:

```
./watsup ./tutorial/*.watsup --prose
```

will generate prose, using every `./watsup` file in directory `./tutorial` as input. There are various options including `--interpreter`, `--latex`, or `--print-il`. You can see all possible options with command `./watsup -help`.

2 Basic Building Blocks of Wasm-DSL

First, we'll study basic building blocks of Wasm-DSL, and how to use them.
For now, You don't have to understand this part completely. You may start from Section 3, and refer to this section when you want to.

2.1 Syntax Definitions

Syntax definitions describe the grammar of the input language or auxiliary constructs. These are essentially type definitions for Wasm-DSL.

```
syntax <name_of_syntax> = <case> | ... | <case>
```

Defining syntax in Wasm-DSL is basically done as above. Use keyword `syntax`, write the name of syntax in lowercase, and simply list the possible cases with the separator `|`. Here, each of the case can be a nonterminal node (usually written in lowercases) which refers to another syntax, or a terminal node (usually written in uppercases).

2.2 Variable Declarations

Variable declarations ascribe the syntactic class (i.e., type) that meta variables used in rules range over.

```
var <name_of_var> : <type>
```

We can explicitly declare a metavariable as above. Use keyword var, and give its name and type. Here, <type> may be a complex form, which contains iteration, parametric syntax, etc.
Also, every syntax name is implicitly usable as a variable of the respective type.

2.3 Relation Declarations

Relation declarations, defining the shape of judgement forms, such as typing or reduction relations. These are essentially type declarations for the meta language. For example:

```
relation Instr_ok: context |- instr : functype
relation Step: config ~> config
```

2.4 Rule Definitions

```
rule <name_of_relation>/<name_of_rule>:
  <content_of_rule>
```

We can define the individual rules for each relation as above. <content_of_rule> has different form, depending on its relation. Every rule is named, so that it can be referenced. Each premise is introduced by a dash and includes the name of the relation it is referencing, easing checking and processing.

2.5 Auxiliary Functions

```
def $<func_name>(<type_of_arg1>, <type_of_arg2>, ... , <type_of_argn>) : <result_type>
def $<func_name>(<arg1>, <arg2>, ... , <argn>) = <result>
```

We can declare auxiliary functions as above. The first line indicates the type of arguments and result of the function. The second line is function body, which describes how the actual result comes out.

3 Syntax

Now, we will start from writing the syntax of Mini-Wasm. Abstract syntax of Mini-Wasm is as follows:

$$\begin{array}{rcl} N & ::= & \mathbb{N} \\ n & ::= & \mathbb{N} \end{array}$$

$$\begin{array}{lrcl} \text{(integer)} & iN(N) & ::= & 0 \mid \dots \mid 2^N - 1 \end{array}$$

$$\begin{array}{lrcl} \text{(character)} & \mathit{char} & ::= & \text{U+00} \mid \dots \mid \text{U+D7FF} \mid \text{U+E000} \mid \dots \mid \text{U+10FFFF} \\ \text{(name)} & \mathit{name} & ::= & \mathit{char}^* \end{array}$$

$$\begin{array}{lrcl} \text{(index)} & \mathit{idx} & ::= & iN(32) \\ \text{(type index)} & \mathit{typeidx} & ::= & \mathit{idx} \\ \text{(function index)} & \mathit{funcidx} & ::= & \mathit{idx} \\ \text{(label index)} & \mathit{labelidx} & ::= & \mathit{idx} \\ \text{(local index)} & \mathit{localidx} & ::= & \mathit{idx} \end{array}$$

$$\text{(number type)} \quad \mathit{valtype} \quad ::= \quad \mathsf{i32} \mid \mathsf{i64}$$

$$\begin{array}{lrcl}
\text{(function type)} & \mathit{functype} & ::= & \mathit{valtype}^* \to \mathit{valtype}^* \\
\text{(external type)} & \mathit{externtype} & ::= & \mathsf{func}\ \mathit{functype}
\end{array}$$

$$\begin{array}{rcl}
\mathrm{size}(\mathsf{i32}) & = & 32 \\
\mathrm{size}(\mathsf{i64}) & = & 64
\end{array}$$

$$\mathit{num}_{\mathit{valtype}} \quad ::= \quad \mathit{iN}(\mathrm{size}(\mathit{valtype}))$$

$$\mathit{binop} \quad ::= \quad \mathsf{add} \mid \mathsf{sub} \mid \mathsf{mul} \mid \mathsf{div}$$

$$\begin{array}{lrcl}
\text{(instruction)} & \mathit{instr} & ::= & \mathsf{nop} \\
 & & \mid & \mathsf{drop} \\
 & & \mid & \mathsf{select} \\
 & & \mid & \mathsf{block}\ \mathit{functype}\ \mathit{instr}^* \\
 & & \mid & \mathsf{loop}\ \mathit{functype}\ \mathit{instr}^* \\
 & & \mid & \mathsf{if}\ \mathit{functype}\ \mathit{instr}^*\ \mathsf{else}\ \mathit{instr}^* \\
 & & \mid & \mathsf{br}\ \mathit{labelidx} \\
 & & \mid & \mathsf{br_if}\ \mathit{labelidx} \\
 & & \mid & \mathsf{call}\ \mathit{funcidx} \\
 & & \mid & \mathsf{return} \\
 & & \mid & \mathsf{const}\ \mathit{valtype}\ \mathit{num}_{\mathit{valtype}} \\
 & & \mid & \mathsf{binop}\ \mathit{valtype}\ \mathit{binop} \\
 & & \mid & \mathsf{local.get}\ \mathit{localidx} \\
 & & \mid & \mathsf{local.set}\ \mathit{localidx}
\end{array}$$

$$\text{(expression)} \quad \mathit{expr} \quad ::= \quad \mathit{instr}^*$$

$$\begin{array}{lrcl}
\text{(type)} & \mathit{type} & ::= & \mathsf{type}\ \mathit{functype} \\
\text{(local)} & \mathit{local} & ::= & \mathsf{local}\ \mathit{valtype} \\
\text{(function)} & \mathit{func} & ::= & \mathsf{func}\ \mathit{typeidx}\ \mathit{local}^*\ \mathit{expr} \\
\text{(external index)} & \mathit{externidx} & ::= & \mathsf{func}\ \mathit{funcidx} \\
\text{(export)} & \mathit{export} & ::= & \mathsf{export}\ \mathit{name}\ \mathit{externidx} \\
\text{(module)} & \mathit{module} & ::= & \mathsf{module}\ \mathit{type}^*\ \mathit{func}^*\ \mathit{export}^*
\end{array}$$

$$\begin{array}{lrcl}
\text{(address)} & \mathit{addr} & ::= & \mathbb{N} \\
\text{(function address)} & \mathit{funcaddr} & ::= & \mathit{addr}
\end{array}$$

$$\text{(value)} \quad \mathit{val} \quad ::= \quad \mathsf{const}\ \mathit{valtype}\ \mathit{num}_{\mathit{valtype}}$$

$$\text{(external value)} \quad \mathit{externval} \quad ::= \quad \mathsf{func}\ \mathit{funcaddr}$$

$$
\begin{array}{llcl}
\text{(function instance)} & \mathit{funcinst} & ::= & \{\mathsf{type}\ \mathit{functype}, \\
 & & & \ \mathsf{module}\ \mathit{moduleinst}, \\
 & & & \ \mathsf{code}\ \mathit{func}\} \\
\text{(export instance)} & \mathit{exportinst} & ::= & \{\mathsf{name}\ \mathit{name}, \\
 & & & \ \mathsf{value}\ \mathit{externval}\} \\
\text{(module instance)} & \mathit{moduleinst} & ::= & \{\mathsf{types}\ \mathit{functype}^*, \\
 & & & \ \mathsf{funcs}\ \mathit{funcaddr}^*, \\
 & & & \ \mathsf{exports}\ \mathit{exportinst}^*\}
\end{array}
$$

$$
\begin{array}{llcl}
\text{(store)} & \mathit{store} & ::= & \{\mathsf{funcs}\ \mathit{funcinst}^*\} \\
\text{(frame)} & \mathit{frame} & ::= & \{\mathsf{locals}\ \mathit{val}^*, \\
 & & & \ \mathsf{module}\ \mathit{moduleinst}\} \\
\text{(state)} & \mathit{state} & ::= & \mathit{store}; \mathit{frame} \\
\text{(configuration)} & \mathit{config} & ::= & \mathit{state}; \mathit{admininstr}^*
\end{array}
$$

$$
\begin{array}{llcl}
\text{(administrative instruction)} & \mathit{admininstr} & ::= & \mathit{instr} \\
 & & | & \mathsf{call\_addr}\ \mathit{funcaddr} \\
 & & | & \mathsf{label}_n\ \{\mathit{instr}^*\}\ \mathit{admininstr}^* \\
 & & | & \mathsf{frame}_n\ \{\mathit{frame}\}\ \mathit{admininstr}^* \\
 & & | & \mathsf{trap}
\end{array}
$$

$$\mathrm{funcaddr}((s; f)) \quad = \quad f.\mathsf{module}.\mathsf{funcs}$$

$$\mathrm{local}((s; f), x) \quad = \quad f.\mathsf{locals}[x]$$

$$\mathrm{with}_{local}((s; f), x, v) \quad = \quad s; f[.\mathsf{locals}[x] = v]$$

$$\mathrm{funcinst}((s; f)) \quad = \quad s.\mathsf{funcs}$$

$$
\begin{array}{lcl}
\mathrm{default}_{\mathsf{i32}} & = & (\mathsf{const}\ \mathsf{i32}\ 0) \\
\mathrm{default}_{\mathsf{i64}} & = & (\mathsf{const}\ \mathsf{i64}\ 0)
\end{array}
$$

Make a new file `1-syntax.wastup`.
We'll declare each of the syntax one by one.

3.1 N, n

$$
\begin{array}{lcl}
N & ::= & \mathbb{N} \\
n & ::= & \mathbb{N}
\end{array}
$$

This syntax is simply written in Wasm-DSL like this:

```
syntax N = nat
syntax n = nat
```

Here, `nat` is a pre-defined syntax, which indicates any natural number. This means the syntax `N` and `n` is a natural number.

3.2 iN

$$\text{(integer)} \quad iN(N) \quad ::= \quad 0 \mid \ldots \mid 2^N - 1$$

```
syntax iN(N) = 0 | ... | 2^N-1
```

Now, the syntax `iN` is declared in regard with parameter `N`. Also, you can use `...` to indicate a range.

3.3 $char$, $name$

$$\begin{array}{llll} \text{(character)} & char & ::= & \text{U+00} \mid \ldots \mid \text{U+D7FF} \mid \text{U+E000} \mid \ldots \mid \text{U+10FFFF} \\ \text{(name)} & name & ::= & char^* \end{array}$$

We need a declaration of *name*, which indicates for a general string. We can declare a syntax for a character like this:

```
syntax char = U+0000 | ... | U+D7FF | U+E000 | ... | U+10FFFF
```

This syntax is built-in, and denotes Unicode code points. Now, `name` is simply an iteration of `char`s. Use `*` to represent a sequence.

```
syntax name = char*
```

3.4 idx, $typeidx$, $funcidx$, $labelidx$, $localidx$

$$\begin{array}{lrll} \text{(index)} & idx & ::= & iN(32) \\ \text{(type index)} & typeidx & ::= & idx \\ \text{(function index)} & funcidx & ::= & idx \\ \text{(label index)} & labelidx & ::= & idx \\ \text{(local index)} & localidx & ::= & idx \end{array}$$

```
syntax idx = I32
```

Since we have four types of index (which are semantically same, but syntactically different), write like this:

```
syntax typeidx = idx
syntax funcidx = idx
syntax labelidx = idx
syntax localidx = idx
```

3.5 $valtype$

$$\text{(number type)} \quad valtype \quad ::= \quad \mathsf{i32} \mid \mathsf{i64}$$

```
syntax valtype = I32 | I64
```

3.6 $functype$

$$\text{(function type)} \quad functype \quad ::= \quad valtype^* \to valtype^*$$

We can use `->` to indicate a function:

```
syntax functype = valtype* -> valtype*
```

3.7 *externtype*

$$\text{(external type)} \quad \mathit{externtype} ::= \mathsf{func}\ \mathit{functype}$$

```
syntax externtype = FUNC functype
```

3.8 *size*, $num_{valtype}$

$$\begin{array}{rcl} \mathrm{size}(\mathsf{i32}) & = & 32 \\ \mathrm{size}(\mathsf{i64}) & = & 64 \end{array}$$

$$\mathit{num}_{\mathit{valtype}} ::= \mathit{iN}(\mathrm{size}(\mathit{valtype}))$$

```
syntax num_(valtype) = iN($size(valtype))
```

The first line declares the syntax `num_`, and the remaining lines give the actual definition. We'll redefine this syntax later on Section 4.2, using function `size`.

3.9 *binop*

$$\mathit{binop} ::= \mathsf{add} \mid \mathsf{sub} \mid \mathsf{mul} \mid \mathsf{div}$$

```
syntax binop = ADD | SUB | MUL | DIV
```

3.10 *instr*

$$\begin{array}{llll} \text{(instruction)} & \mathit{instr} & ::= & \mathsf{nop} \\ & & | & \mathsf{drop} \\ & & | & \mathsf{select} \\ & & | & \mathsf{block}\ \mathit{functype}\ \mathit{instr}^* \\ & & | & \mathsf{loop}\ \mathit{functype}\ \mathit{instr}^* \\ & & | & \mathsf{if}\ \mathit{functype}\ \mathit{instr}^*\ \mathsf{else}\ \mathit{instr}^* \\ & & | & \mathsf{br}\ \mathit{labelidx} \\ & & | & \mathsf{br\_if}\ \mathit{labelidx} \\ & & | & \mathsf{call}\ \mathit{funcidx} \\ & & | & \mathsf{return} \\ & & | & \mathit{valtype}.\mathsf{const}\ \mathit{num}_{\mathit{valtype}} \\ & & | & \mathit{valtype}.\mathit{binop} \\ & & | & \mathsf{local.get}\ \mathit{localidx} \\ & & | & \mathsf{local.set}\ \mathit{localidx} \end{array}$$

```
syntax instr =
  | NOP
  | DROP
  | SELECT
  | BLOCK functype instr*
  | LOOP functype instr*
  | IF functype instr* ELSE instr*
  | BR labelidx
  | BR_IF labelidx
  | CALL funcidx
  | RETURN
  | CONST valtype num_(valtype)
  | BINOP valtype binop
  | LOCAL.GET localidx
  | LOCAL.SET localidx
```

3.11 *expr*

$$\text{(expression)} \quad \mathit{expr} \quad ::= \quad \mathit{instr}^*$$

```
syntax expr = instr*
```

3.12 *module*

$$\begin{array}{llcl}
\text{(type)} & \mathit{type} & ::= & \mathsf{type}\ \mathit{functype} \\
\text{(local)} & \mathit{local} & ::= & \mathsf{local}\ \mathit{valtype} \\
\text{(function)} & \mathit{func} & ::= & \mathsf{func}\ \mathit{typeidx}\ \mathit{local}^*\ \mathit{expr} \\
\text{(external index)} & \mathit{externidx} & ::= & \mathsf{func}\ \mathit{funcidx} \\
\text{(export)} & \mathit{export} & ::= & \mathsf{export}\ \mathit{name}\ \mathit{externidx} \\
\text{(module)} & \mathit{module} & ::= & \mathsf{module}\ \mathit{type}^*\ \mathit{func}^*\ \mathit{export}^*
\end{array}$$

Declare *module* and its subcomponents as follows:

```
syntax type = TYPE functype
syntax local = LOCAL valtype
syntax func = FUNC typeidx local* expr
syntax externidx = FUNC funcidx
syntax export = EXPORT name externidx
syntax module = MODULE type* func* export*
```

3.13 *addr*, *funcaddr*

$$\begin{array}{llcl}
\text{(address)} & \mathit{addr} & ::= & \mathbb{N} \\
\text{(function address)} & \mathit{funcaddr} & ::= & \mathit{addr}
\end{array}$$

```
syntax addr = nat
syntax funcaddr = addr
```

3.14 *val*

$$\text{(value)} \quad \mathit{val} \quad ::= \quad \mathsf{const}\ \mathit{valtype}\ \mathit{num}_{\mathit{valtype}}$$

```
syntax val = CONST valtype num_(valtype)
```

3.15 *externval*

$$\text{(external value)} \quad externval \quad ::= \quad \mathsf{func}\ funcaddr$$

```
syntax externval = FUNC funcaddr
```

3.16 *funcinst*, *exportinst*, *moduleinst*

$$\begin{array}{lrcl}
\text{(function instance)} & funcinst & ::= & \{\mathsf{type}\ functype, \\
 & & & \ \mathsf{module}\ moduleinst, \\
 & & & \ \mathsf{code}\ func\} \\
\text{(export instance)} & exportinst & ::= & \{\mathsf{name}\ name, \\
 & & & \ \mathsf{value}\ externval\} \\
\text{(module instance)} & moduleinst & ::= & \{\mathsf{types}\ functype^*, \\
 & & & \ \mathsf{funcs}\ funcaddr^*, \\
 & & & \ \mathsf{exports}\ exportinst^*\}
\end{array}$$

```
syntax funcinst =
{ TYPE functype,
        MODULE moduleinst,
        CODE func }
syntax exportinst =
{ NAME name,
        VALUE externval }

syntax moduleinst =
{ TYPES functype*,
        FUNCS funcaddr*,
        EXPORTS exportinst* }
```

3.17 *store*, *frame*, *state*, *config*

$$\begin{array}{lrcl}
\text{(store)} & store & ::= & \{\mathsf{funcs}\ funcinst^*\} \\
\text{(frame)} & frame & ::= & \{\mathsf{locals}\ val^*, \\
 & & & \ \mathsf{module}\ moduleinst\} \\
\text{(state)} & state & ::= & store; frame \\
\text{(configuration)} & config & ::= & state; admininstr^*
\end{array}$$

```
syntax store =
{ FUNCS funcinst* }

syntax frame =
{ LOCALS val*,
        MODULE moduleinst }

syntax state = store; frame
syntax config = state; admininstr*
```

3.18 *admininstr*

$$\begin{array}{llcl}
\text{(administrative instruction)} & \mathit{admininstr} & ::= & \mathit{instr} \\
& & | & \mathsf{call_addr}~\mathit{funcaddr} \\
& & | & \mathsf{label}_{n}~\{\mathit{instr}^*\}~\mathit{admininstr}^* \\
& & | & \mathsf{frame}_{n}~\{\mathit{frame}\}~\mathit{admininstr}^* \\
& & | & \mathsf{trap}
\end{array}$$

```
syntax admininstr =
| instr
| CALL_ADDR funcaddr
| LABEL_ n `{instr*} admininstr*
| FRAME_ n `{frame} admininstr*
| TRAP
```

3.19 funcaddr

$$\mathrm{funcaddr}((s; f)) \quad = \quad f.\mathsf{module}.\mathsf{funcs}$$

```
def $funcaddr(state) : funcaddr*
def $funcaddr((s; f)) = f.MODULE.FUNCS
```

3.20 local

$$\mathrm{local}((s; f), x) \quad = \quad f.\mathsf{locals}[x]$$

```
def $local(state, localidx) : val
def $local((s; f), x) = f.LOCALS[x]
```

3.21 $\mathrm{with}_{\mathit{local}}$

$$\mathrm{with}_{\mathit{local}}((s; f), x, v) \quad = \quad s; f[.\mathsf{locals}[x] = v]$$

```
def $with_local(state, localidx, val) : state
def $with_local((s; f), x, v) = s; f[.LOCALS[x] = v]
```

3.22 funcinst

$$\mathrm{funcinst}((s; f)) \quad = \quad s.\mathsf{funcs}$$

```
def $funcinst(state) : funcinst*
def $funcinst((s; f)) = s.FUNCS
```

3.23 $\mathrm{default}_{\mathit{valtype}}$

$$\begin{array}{lcl}
\mathrm{default}_{\mathsf{i32}} & = & (\mathsf{const}~\mathsf{i32}~0) \\
\mathrm{default}_{\mathsf{i64}} & = & (\mathsf{const}~\mathsf{i64}~0)
\end{array}$$

```
def $default_(valtype) : val
def $default_(I32) = (CONST I32 0)
def $default_(I64) = (CONST I64 0)
```

4 Metavariables

We're done with writing syntax of Mini-Wasm. Now, we will declare metavariables, which is used in reduction rules.
There are three ways to make use of metavariables.

4.1 Explicit Declarartion

Declare variables in file `1-syntax.wastup` as follows:

```
var x : idx
var l : labelidx
var t : valtype
var ft : functype
var in : instr
var e : instr*
var ty : type
var loc : local
var ex : export
var st : start
```

4.2 Using Syntax Name

Also, we can use the syntax name directly as a variable of same type. For example, instead of below code:

```
rule Step/example:
  z; (CONST t c_1) (CONST t c_2)  ~>  z; (CONST t c_1)
```

we can write like this:

```
rule Step/example:
  z; (CONST valtype c_1) (CONST valtype c_2)  ~>  z; (CONST valtype c_1)
```

4.3 Using Without Declaration

Finally, we can use a metavariable without declaration in suitable situations. In this case, its type will be inferred. Now, we may rewrite the above rule as:

```
rule Step/example:
  z; (CONST anyname c_1) (CONST anyname c_2)  ~>  z; (CONST anyname c_1)
```

5 Functions

5.1 Declaring Function Body

Function body can be multiple cases, which defines a function by pattern matching.
For example, we can define a function `size`, which returns the size of a `valtype` as:

```
def $size(valtype) : nat
def $size(I32) = 32
def $size(I64) = 64
```

Now, the function `size` will return `32` for input `I32`, and `64` for input `I64`.

We can also add condition to a function body.
For example, we can define a function `min`, which returns a smaller integer between two:

```
def $min(nat, nat) : nat
def $min(i, j) = i
  -- if i < j
def $min(i, j) = j
```

Now, the function `min` will return `i` if `i < j`, else `j`.

5.2 Using Functions

Functions can be used in definition of other functions or reduction rules. When using function, write as below form:

```
$<func_name>(<arg1>, <arg2>, ... , <argn>)
```

For example, put the definition of the function `size` before the syntax `num_` from Section 2.7. Then, we can use `size` to redefine `num_` as below:

```
syntax num_(valtype) = iN($size(valtype))
```

6 Reduction Rules

Now, let's write the reduction rules for the instructions. First, make a new file `8-reduction.wastup` and write as follows:

```
relation Step: config ~> config
```

It means that `rules` of `relation Step` receives `config` as its input and then yields new `config` as its output. Here, `config` is defined in file `4-runtime.watsup`:

```
syntax config = state; admininstr*
syntax admininstr =
  | instr
  | CALL_ADDR funcaddr
  | LABEL_ n ‘{instr*} admininstr*
  | FRAME_ n ‘{frame} admininstr*
  | TRAP
```

As you can see, `config` is a pair of `state` and sequence of `admininstr`. `admininstr` is a superset of `instr`, which contains some additional administrative instructions.
Declaring reduction rule in Wasm-DSL is basically done like this:

```
rule Step/<rule name>:
  <input state>; <input instructions>  ~>  <output state>; <output instructions>
  -- if <condition>
  ...
  -- if <condition>
```

This means that if every `condition` is satisfied, then the state and instructions from stack is reduced to the right hand side. There may be no conditions.
Now we'll declare each of the reduction rules one by one.

6.1 NOP

Use `eps` to indicate an empty instruction sequence.
Use metavariable `z` for a `state`.

Answer:

```
rule Step/nop:
  z; NOP  ~>  z; eps
```

6.2 DROP

Use following syatax `val`, which is defined in `4-runtime.watsup`:

```
syntax val = CONST valtype val_(valtype)
```

as a variable.

Answer:

```
rule Step/drop:
  z; val DROP  ~>  z; eps
```

6.3 SELECT

When we get `SELECT` from stack, we have two cases: condition is true or false.
Define each of the case as seperate rule as follows:

```
rule Step/select-true:
  ...

rule Step/select-false:
  ...
```

Write like `val_1`, `val_2` (and so on) to distinguish multiple `val`s.
Use `=/=` and `=` for integer comparison.

Answer:

```
rule Step/select-true:
  z; val_1 val_2 (CONST I32 c) SELECT  ~>  z; val_1
  -- if c =/= 0

rule Step/select-false:
  z; val_1 val_2 (CONST I32 c) SELECT  ~>  z; val_2
  -- if c = 0
```

As you can see here, you can distinguish multiple metavariables with same type by adding _1, _2 (and so on) after the metavariable name. Also, the names of two rules should be the same before hyphen (-), so that they can be prosed as the same reduction rule.

6.4 BINOP

When we get BINOP from stack, there are three cases for binop_(valtype): ADD, SUB, MUL.
Define each of the case as seperate rule, and use built-in functions **iadd**, **isub**, **imul** as follows:

```
def $iadd(N, iN(N), iN(N)) : iN(N)
def $isub(N, iN(N), iN(N)) : iN(N)
def $imul(N, iN(N), iN(N)) : iN(N)
def $iadd(N, c_1, c_2) = $((c_1 + c_2) \ 2^N)
def $isub(N, c_1, c_2) = $((c_1 - c_2) \ 2^N)
def $imul(N, c_1, c_2) = $((c_1 * c_2) \ 2^N)

rule Step/binop-add:
  ...

rule Step/binop-sub:
  ...

rule Step/binop-mul:
  ...
```

Use function **size**, which is defined from subsection 4.1.

Answer:

```
def $iadd(N, iN(N), iN(N)) : iN(N)
def $isub(N, iN(N), iN(N)) : iN(N)
def $imul(N, iN(N), iN(N)) : iN(N)

rule Step/binop-add:
  z; (CONST t c_1) (CONST t c_2) (BINOP t ADD)  ~>  z; (CONST t c)
-- if $iadd($size(t), c_1, c_2) = c

rule Step/binop-sub:
z; (CONST t c_1) (CONST t c_2) (BINOP t SUB)  ~>  z; (CONST t c)
-- if $isub($size(t), c_1, c_2) = c

rule Step/binop-mul:
z; (CONST t c_1) (CONST t c_2) (BINOP t MUL)  ~>  z; (CONST t c)
-- if $imul($size(t), c_1, c_2) = c
```

Now, we may combine them by declaring a new function binop as follows:

```
def $iadd(N, iN(N), iN(N)) : iN(N)
def $isub(N, iN(N), iN(N)) : iN(N)
def $imul(N, iN(N), iN(N)) : iN(N)

def $binop(valtype, binop_(valtype), val_(valtype), val_(valtype)) : val_(valtype)*
def $binop(t, ADD, c_1, c_2) = $iadd($size(t), c_1, c_2)
def $binop(t, SUB, c_1, c_2) = $isub($size(t), c_1, c_2)
def $binop(t, MUL, c_1, c_2) = $imul($size(t), c_1, c_2)

rule Step/binop:
  z; (CONST t c_1) (CONST t c_2) (BINOP t binop)  ~>  z; (CONST t c)
  -- if $binop(t, binop, c_1, c_2) = c
```

6.5 BLOCK

To indicate a label, refer to following syntax:

```
LABEL_ n `{instr*} admininstr*
```

Answer:

```
rule Step/block-eps:
  z; (BLOCK eps instr*)  ~>  z; (LABEL_ 0 `{eps} instr*)
rule Step/block-val:
  z; (BLOCK t instr*)  ~>  z; (LABEL_ 1 `{eps} instr*)
```

Here, you may combine them by using `/\` for 'and', `\/` for 'or', and `?` for an optional argument:

```
rule Step/block:
  z; (BLOCK t? instr*)  ~>  z; (LABEL_ n `{eps} instr*)
  -- if t? = eps /\ n = 0 \/ t? =/= eps /\ n = 1
```

6.6 LOOP

Answer:

```
rule Step/loop:
  z; (LOOP t? instr*)  ~>  z; (LABEL_ 0 `{LOOP t? instr*} instr*)
```

6.7 IF

Answer:

```
rule Step/if-true:
z; (CONST I32 c) (IF t? instr_1* ELSE instr_2*)  ~>  z; (BLOCK t? instr_1*)
-- if c =/= 0

rule Step/if-false:
z; (CONST I32 c) (IF t? instr_1* ELSE instr_2*)  ~>  z; (BLOCK t? instr_2*)
-- if c = 0
```

6.8 BR

For arithmetic expressions, you should write like `$(l+1)`, instead of `l+1`.
Use `^` for a sequence with given length. e.g. `instr^n`.

Answer:

```
rule Step/br-zero:
z; (LABEL_ n `{instr'*} val'* val^n (BR 0) instr*)  ~>  z; val^n instr'*

rule Step/br-succ:
z; (LABEL_ n `{instr'*} val* (BR $(l+1)) instr*)  ~>  z; val* (BR l)
```

6.9 BR_IF

Answer:

```
rule Step/br_if-true:
  z; (CONST I32 c) (BR_IF l)  ~>  z; (BR l)
  -- if c =/= 0

rule Step/br_if-false:
  z; (CONST I32 c) (BR_IF l)  ~>  z; eps
  -- if c = 0
```

6.10 CALL

To get a sequence of `funcaddr` from `state z`, write `$funcaddr(z)`.
To get `nth` element of sequence `seq`, write `seq[n]`.

Answer:

```
rule Step/call:
  z; (CALL x)  ~>  z; (CALL_ADDR $funcaddr(z)[x])
```

6.11 FRAME

To indicate a frame, refer to following syntax:

```
FRAME_ n `{frame} admininstr*
```

Answer:

```
rule Step/frame-vals:
  z; (FRAME_ n `{f} val^n)  ~>  z; val^n
```

6.12 RETURN

Answer:

```
rule Step/return-frame:
  z; (FRAME_ n `{f} val'* val^n RETURN instr*)  ~>  z; val^n

rule Step/return-label:
  z; (LABEL_ n `{instr'*} val* RETURN instr*)  ~>  z; val* RETURN
```

6.13 LOCAL.GET

To get a local value from `state z` and `idx x`, write `$local(z, x)`.

Answer:

```
rule Step/local.get:
  z; (LOCAL.GET x)  ~>  z; $local(z, x)
```

6.14 LOCAL.SET

To get a new state which is exactly same with `state z` except that its local value with `idx x` is `val v`, write `$with_local(z, x, v)`.

Answer:

```
rule Step/local.set:
  z; val (LOCAL.SET x)  ~>  $with_local(z, x, val); eps
```